Paper 048-30

%CYARROW, A NEW ANNOTATE MACRO

Chang Y. Chung, Princeton University, Princeton, NJ
Ya Huang, Amylin Pharmaceuticals, Inc., San Diego, CA

ABSTRACT

We present a macro, %CYARROW, that makes it easy to draw arrows using SAS®/GRAPH's superior Annotate facility. %CYARROW is specifically designed to be used in conjunction with %ANNOMAC macros, a set of SAS macros that provides a shortcut when creating an Annotate dataset. %CYARROW is an alternative to the undocumented (and "broken") %ARROW macro. In addition to honoring the five standard options for the DRAW Annotate function (COLOR, HSYS, LINE, SIZE, and WHEN), %CYARROW offers sophisticated arrow shape options that control the length and the angle of the arrow tip and its aspect ratio.

ANNOTATE DATA SET AND ANNOTATE MACROS

The Annotate facility is a part of SAS/GRAPH. It is most commonly used to enhance graphics output by adding text and other graphic elements to them. It also creates advanced, custom graphics output from scratch in conjunction with PROC GANNO.

In order to use the Annotate facility, one creates a dataset (an Annotate dataset) with specific variables and values. The Annotate dataset then is fed into a graphics procedure. The procedure, in turn, processes each observation of the Annotate dataset as an instruction to the facility. A good introduction to the Annotate facility is the, excellent, short volume by Carpenter(1999a).

Since the bulk of work in using Annotate facility lies in creating the Annotate dataset, SAS kindly provides a set of macros, %ANNOMAC, that significantly reduces the amount of work. As Rorie and Duncan(2003) demonstrate (also see Carpenter(1999b)), a single %LINE( ) Annotate macro call can replace: (1) creating two observations; (2) assigning the function variable values (MOVE, for the first observation; and DRAW for the second); (3) assigning x and y coordinates for the starting and ending points of a line; and (4) assigning color, line type, and thickness.

The %ANNOMAC includes %BAR to create a fillable rectangle, %LINE to draw a line, %SLICE to draw a pie slice, and %LABEL to write text at the specified location, to name a few. Our installation of SAS Release 9.1.2 includes the source code file for the %ANNOMAC macros (in the "SAS 9.1/core/sasmacro" directory). To our delight, the source code includes an undocumented macro called %ARROW.

BROKEN %ARROW

When we try to use the %ARROW macro, however, we come to an understanding of the reason why it is not documented — it is broken. Disappointed, but motivated, we wrote our own Annotate macro, %CYARROW.

```
Filename gout "%sysfunc(pathname(WORK))/broken.emf";

goptions reset=all
  device=emf gsfname=gout gsfmode=replace
  hpos=40 vpos=40 hsize=2in vsize=2in
;

data anno;
  %annomac(nomsg)
  %dclanno
  %frame(CXFF0000, 1, 1)
  %arrow(10,10,30,30,6,black,1)
run;

proc ganno annotate=anno;
run;
quit;
```



%CYARROW

The new macro draws a very simple arrow, but does it well. Here is an example.

```
filename gout "%sysfunc(pathname(WORK))/cy1.emf";

goptions reset=all
  device=emf gsfname=gout gsfmode=replace
  hpos=40 vpos=40 hsize=2in vsize=2in
;

data anno;
  %annomac(nomsg)
  %dclanno
  %frame(CXFF0000, 1, 1)
  radius = 16;
  do deg = 0 to 360 by 10;
    rad = deg * constant('pi') / 180;
    cx = 20 + radius * cos(rad);
    cy = 20 + radius * sin(rad);
    %cyArrow(25,22,cx,cy,color="cx0000ff")
  end;
run;

proc ganno annotate=anno;
run;
quit;
```



HOW TO USE

%CYARROW is designed to be used with the %ANNOMAC macros. The macro follows the %ANNOMAC convention in macro implementation and usage. For instance, it assumes that users issue %DCLANNO macro to set up required variables, just as other Annotate macros do.

In terms of implementation, %CYARROW features much improved internals: it has robust input parameter checking and error handling routines, and it minimizes unwanted side-effects using stacks. Still, it is very simple to use.

For example, the following short line draws an arrow.

```
%cyArrow(1,1,3,3) /* draw an arrow starting from (x,y)=(1,1), pointing to (3,3) */
```

There are two sets of options. The first set of five options (COLOR, LINE, SIZE, HSYS, and WHEN) are the same as the options for the DRAW and other Annotate functions.

COLOR controls the color of the arrow. It can take a string literal or a character variable of length $8.

```
%cyArrow(x1, y1, x2, y2, color = “cx0000ff”) /* blue */
```

LINE specifies the line type – values range from 0 (no line) to 46 (three dots and dashes). SIZE controls the thickness of the line, the unit of which depends on HSYS. For example, when HSYS=″4″, then LINE=1 indicates that the arrow lines should be one cell unit thick. %CYARROW inherits the HSYS value that is assigned before the %CYARROW call (presumably using %DECLANNO or %SYSTEM Annotate macros). But you can temporarily override the unit with HSYS= option.

The WHEN= option controls whether the arrow should be drawn before (“B″) or after (“A″) the graphs, when the Annotate dataset is used with SAS/GRAPH procedures such as PROC GPLOT.

The second set of options control how the tip of the arrow should be drawn. An arrow has a “shaft”, and two “barbs,” as shown below.




```
%include "cyArrow.sas";

data anno;
  %annomac
  %dclanno
  length text $20;
  %system(4,4,4)

  /* grid */
  drop i;
  do i = 0 to 100 by 10;

      %line(i,0, i,100, cxe0e0e0, 1, 1)
      %line(0,i, 100,i, cxe0e0e0, 1, 1)
  end;

  /* the large arrow */
  %cyArrow(20, 20, 70, 70, color="cx0000ff", size=5, barbLength=0.2, barbAngle=45)

  /* an arrow pointing "shaft" */
  %cyArrow(50, 20, 40.5, 39.5, color="cx000000", size=0.05)
  %label(  50, 20, "Shaft", cx000000, 0, 0, 7, Simplex, 6)

  /* arrows pointing "barbs" */
  %cyArrow(85, 45, 65, 69, color="cx000000", size=0.05)
  %cyArrow(85, 45, 71, 64, color="cx000000", size=0.05)
  %label(  85, 45, "Barbs", cx000000, 0, 0, 7, Simplex, 5)

  /* barbLength */
  drop j;
  j = 70 - sqrt(2) * 10;
  %line(j,  71.5, 70, 71.5, cx000000, 1, 0.01)
  %line(j,  70,    j, 73,   cx000000, 1, 0.01)
  %line(70, 70,   70, 73,   cx000000, 1, 0.01)
  %cyArrow(65, 90, 65, 71.5, color="cx000000", size=0.05)
  %label(  65, 90, "BarbLength", cx000000, 0, 0, 7, Simplex, B)


  /* barbAngle */
  when = "A";

  %slice(  70, 70, 180, 45, 13, cx000000, empty, 0)
  %cyArrow(35, 65, 57,  65, color="cx000000", size=0.05)
```

```
  %label(  35, 65, "BarbAngle", cx000000, 0, 0, 7, Simplex, A)
run;

dm 'graph1; end;' wedit;
filename gout "shaftAndBarb.emf";
goptions reset=all vsize=4in hsize=4in vpos=100 hpos=100
  device=emf gsfname=gout gsfmode=replace
  targetdevice=emf goutmode=replace
;

proc ganno annotate=anno;
run;
/*
data one;
  retain one_y one_x .;
run;
proc gplot data=one;
  plot one_y * one_x
     / haxis=(0 to 1 by 1) vaxis=(0 to 1 by 1) annotate=anno;
run;
quit;
*/
goptions reset=all;
```

The BARBANGLE option allows users to change the angle (in degrees between 1 and 90) between a barb and the shaft. The default value is 30.

The length of the barbs is controlled by two options: BARBLENGTHTYPE and BARBLENGTH. The BARBLENGTHTYPE can be either F (for Fixed) or P (for Proportional). The default is P, under which the value specified in BARBLENGTH indicates length as a proportion of the shaft's length. The default value is 0.05, meaning 5 percent of the shaft length. When the BARBLENGTHTYPE="F", on the other hand, the BARBLENGTH indicates the absolute length in HSYS unit.

An aspect ratio refers to the ratio of the width(x) to height(y). Since the shaft is already defined by the two end-points, only the barbs are affected by the aspect ratio. The barbs' length and angle are adjusted by BARBASPECTRATIO.



```
%include "cyArrow.sas";

/*-- test data set ------------------------------------------------*/
data one;
  one_x =-10; one_y =-10; output;
  one_x = 10; one_y = 10; output;
run;

data anno;

  %annomac(nomsg)
  %dclanno
  %system(2,2,2)
  %frame(CXFF0000, 1, 1)

  length text $30;
  radius = 3.5;

  %macro doCircle(ar=, labelX=, labelY=
    ,offsetX=, offsetY=, centerX=, centerY=
  );
    function="label";
    text="barbAspectRatio = &ar.";
    font="simplex";
    x=&labelX.;
    y=&labelY.;
    size=0.5;
    output;
    do deg = 0 to 360 by 15;
      rad = deg * constant('pi') / 180;
      xx =  &offsetX. + radius * (1/1.1429) * cos(rad);
      yy =  &offsetY. + radius * (1.1429)   * sin(rad);
      %cyArrow(&CenterX., &CenterY., xx, yy
        , barbLengthType="Fixed"
        , barbLength=0.4, barbAspectRatio=&ar.
        , color="cx0000ff", size=0.5
      )
    end;
  %mend doCircle;
  %doCircle(ar=1.0000, labelX=-5, labelY=9.5
    , offsetX=-4.5, offsetY= 4.5, centerX=-5, centerY= 5)
  %doCircle(ar=0.7000, labelX= 5, labelY=9.5
    , offsetX= 4.5, offsetY= 4.5, centerX= 5, centerY= 5)
  %doCircle(ar=1.1429, labelX= 5, labelY=-9.5
    , offsetX= 4.5, offsetY=-4.5, centerX= 5, centerY=-5)
  %doCircle(ar=2.0000, labelX=-5, labelY=-9.5
    , offsetX=-4.5, offsetY=-4.5, centerX=-5, centerY=-5)
run;

dm log 'graph1; clear; end;' wedit;   /* close the graph1 window */
goptions goutmode=replace; /* entire contents of catalog replaced */

/*-- aspect ratio test - gplots---------------------------------*/
%macro doGPlot(hsize=, vsize=, labelSize=);
  dm 'graph1; end;' wedit;
  filename gout "shaftAndBarb.emf";
  goptions reset=all hsize=&hsize. vsize=&vsize.
    device=emf gsfname=gout gsfmode=replace
    targetdevice=emf goutmode=replace
  ;
  proc gplot data=one;
    title font="Tahoma" "HSize=&hsize. and VSize=&vsize.";
    plot one_y*one_x/ annotate=anno grid;
  run;
  quit;
  title;
  goptions reset=all;
%mend doGPlot;
%doGPlot(hsize=8 in, vsize=7 in)
```

Above shows the effects of the BARBASPECTRATIO. The aspect ratio of the graphic area is 1.1429, since we specify HSIZE = 8IN and VSIZE = 7IN (8 / 7 = 1.1429) in the GOPTIONS statement. With the default

(BARBASPECTRATIO=1), the barbs are not of equal length and the angles between a barb and the shaft are different. Also the amount of "distortion" depends on the length of the arrow and the direction. The same kind of distortion happens when the BARBASPECTRATIO is equal to 0.7 or 2.0. By specifying the BARBASPECTRATIO, which is the same as that of the graphic area (here 1.1429), the arrows look best.

HOW IT WORKS

The source code for the %CYARROW is presented below. The code is commented and formatted to enhance readability.

Early on, we made some design choices. Recognizing that this is a "DATA-step macro" (which is to be called within a DATA step), we decided to minimize unwanted side-effects that might cause problems such as variable and label name collisions. Thus we prefix all the names (including DATA step label, macro label, variables, and macro variables) by utilizing a random name generated for each %CYARROW macro call (see line marked with (1)). At the end of the macro, all the temporary variables are listed in the DROP statement (7).

%CYARROW checks input errors at both macro compile time and DATA step run time. The macro compile time error checks are simply to see if any parameter is empty. Most of the error checks are done at DATA step run time – the DATA step code for error handling is simple enough that we implement it using a parameterized string substitution in lieu of subroutine calls (see (2) and (3)).

Unwanted side effects may be caused by DRAW function, since it leaves behind the last coordinates in the X[Y]LAST variables: XLAST, YLAST, XLASTT, and YLASTT. Since we require both the start and end point coordinates in %CYARROW, there is no need for keeping the last coordinates. So, we preserve the values of these variables the same as before the %CYARROW is called -- by storing the current X[Y]LAST values before using DRAW function and restoring them when we are done with drawing the arrows. The PUSH and POP Annotate functions work very well in this context ((4) and (6)).

In order to ease the maintenance burden, we do not call other Annotate macros inside %CYARROW. This also enables us to write compact DATA step code (5).

```
%macro cyArrow(
                          /* coordinates --------------------------- */

    x1, y1                /* starting point coordnate -- required    */
  , x2, y2                /* end point coordinate     -- required    */


                          /* cyArrow specific options -------------- */

  , barbAngle      =30     /* angle between barb and shaft in degrees */
  , barbLengthType ="P"    /* [F]ixed or [P]roportional to shaft len  */
  , barbLength     =0.05   /* if type=F then absolute length in hsys  */
                           /*   unit. otherwise proportion of shaft   */
                           /*   length                                */
  , barbAspectRatio=1      /* the ratio of width(x) to height(y)      */


                          /* draw function options ----------------- */

  , color          ="*"    /* color codes.                           */
  , hsys           =hsys   /* coord sys for size option              */
  , line           =1      /* line type 1...46                       */
  , size           =0.5    /* thickness of lines in hsys units       */
  , when           ="A"    /* annotate "A"fter gproc outputs         */

);

%*-- draw a simple arrow using annotate facility. This replaces     --*;
%*--   the undocumented macro %arrow()                              --*;
%*-- by chang y chung and ya huang                                  --*;
%*-- v1.0 on 2004-07-26                                            --*;
%*-- v1.5 on 2004-07-26 fixed the exit part                        --*;
```

```
%*-- helpers ------------------------------------------------------*;
     %*-- a random prefix for "global" things ----------------------*;
     %global cyArrow;
     %let cyArrow = %sysfunc(putn(%sysfunc(int(1e8*                                    (1)
          %sysfunc(ranuni(0)))),z8.));

     %*-- for data step runtime error handling ----------------------*;
     %local err cond msg;
     %let err = %nrstr(                                                                (2)
          if %unquote(&cond.) then do;
               put "&preMsg. %unquote(&msg.). &postMsg.";
               function = "comment";
               output;
               goto _&cyArrow._exit;
          end;
     );
     %local preMsg postMsg commentAndExit;
     %let preMsg = NOTE: (cyArrow);
     %let postMsg= arrow not drawn.;



%*-- macro compile time error check ---------------------------------*;

     %*-- no empty parameters. --------------------------------------*;
     %local params param value i;
     %let params = x1 y1 x2 y2 barbAngle barbLength barbLengthType
          barbAspectRatio color hsys line size when;
     %let i = 1;
     %let param = %scan(&params.,&i);
     %do %while (&param.^=);
          %if %superq(&param.)= %then %do;
               %put &preMsg. &param. should not be blank. &postMsg.;
               %goto exit;
          %end;
          %let i = %eval(&i. + 1);
          %let param = %scan(&params., &i.);
     %end;

%*-- data step runtime check ----------------------------------------*;

     %*-- coordinates cannot be missing ----------------------------*;
     %local i coord coords;
     %let coords = x1 y1 x2 y2;
     %do i = 1 %to 4;
          %let coord = %scan(&coords., &i.);                                           (3)
          _&cyArrow._&coord. = %unquote(&&&coord..);
          %let cond = missing(_&cyArrow._&coord.);
          %let msg  = &coord. should not be missing;
          %unquote(&err.)
     %end;

     %*-- barbs ----------------------------------------------------*;
     _&cyArrow._barbAngle = (%unquote(&barbAngle.));
     %let cond = not (0<=_&cyArrow._barbAngle<=90);
     %let msg  = barbAngle should be between 0 to 90 degrees;
     %unquote(&err.)

     _&cyArrow._barbLengthType = upcase(trim(left(
          %unquote(&barbLengthType.))));
     %let cond = missing(_&cyArrow._barbLengthType);
     %let msg  = barbLengthType should not be missing;
     %unquote(&err.)

     _&cyArrow._barbLengthType = substr(_&cyArrow._barbLengthType,1,1);
     %let cond = not (_&cyArrow._barbLengthType in ("P" "F"));
     %let msg  = barbLengthType should be either [P]roportional
          or [F]ixed;
     %unquote(&err.)

     _&cyArrow._barbLength = (%unquote(&barbLength.));
```

```
     %let cond = not (0 <= _&cyArrow._barbLength);
     %let msg  = barbLength should not be negative;
     %unquote(&err.)

     if _&cyArrow._barbLengthType = "P" then do;
          %let cond = not (_&cyArrow._barbLength<=1.0);
          %let msg  = Proportional type barbLength should be/*
               */ between 0 and 1;
          %unquote(&err.)
     end;

     _&cyArrow._barbAspectRatio = (%unquote(&barbAspectRatio.));
     %let cond = not (0 < _&cyArrow._barbAspectRatio);
     %let msg  = barbAspectRatio should be larger than zero;
     %unquote(&err.)


%*-- calculation for the shaft ---------------------------------------*;

     %*-- always adjust y for aspect --------------------------------*;
     _&cyArrow._ay1 = _&cyArrow._y1 * _&cyArrow._barbAspectRatio**-1;
     _&cyArrow._ay2 = _&cyArrow._y2 * _&cyArrow._barbAspectRatio**-1;

     %*-- calculate shaft angle and length --------------------------*;
     _&cyArrow._shaftLength = sqrt(
            (_&cyArrow._x1  - _&cyArrow._x2)**2
          + (_&cyArrow._ay1 - _&cyArrow._ay2)**2
     );
     %*-- check -----------------------------------------------------*;
     %let cond = _&cyArrow._shaftLength <= 0;
     %let msg  = shaft length <= 0;
     %unquote(&err.)

     %*-- direction -------------------------------------------------*;
     _&cyArrow._shaftDirection = atan2(
            _&cyArrow._ay1 - _&cyArrow._ay2
          , _&cyArrow._x1  - _&cyArrow._x2
     );

%*-- calculation for the barbs ---------------------------------------*;

     %*-- angle -----------------------------------------------------*;
     _&cyArrow._barbAngle  =
          _&cyArrow._barbAngle * constant('pi') / 180
     ;
     if _&cyArrow._barbLengthType = "P" then do;
          _&cyArrow._barbLength =
               _&cyArrow._shaftLength * _&cyArrow._barbLength
          ;
     end;
     %*-- check -----------------------------------------------------*;
     %let cond = _&cyArrow._barbLength <= 0;
     %let msg  = barb length <= 0;
     %unquote(&err.)


     %*-- coordinates -----------------------------------------------*;
     _&cyArrow._barbX1 = _&cyArrow._x2 + _&cyArrow._barbLength
          * cos(_&cyArrow._shaftDirection + _&cyArrow._barbAngle);
     _&cyArrow._barbY1 = _&cyArrow._y2 + _&cyArrow._barbLength
          * sin(_&cyArrow._shaftDirection + _&cyArrow._barbAngle)
          * _&cyArrow._barbAspectRatio;
     _&cyArrow._barbX2 = _&cyArrow._x2 + _&cyArrow._barbLength
          * cos(_&cyArrow._shaftDirection - _&cyArrow._barbAngle);
     _&cyArrow._barbY2 = _&cyArrow._y2 + _&cyArrow._barbLength
          * sin(_&cyArrow._shaftDirection - _&cyArrow._barbAngle)
          * (_&cyArrow._barbAspectRatio);


%*-- save xlast, ylast, xlastt, ylastt and other options-------------*;                    (4)
     function = "push";
```

```
     output;
     _&cyArrow._color = trim(color);
     _&cyArrow._line  = 1 * line;
     _&cyArrow._size  = 1 * size;
     _&cyArrow._hsys  = trim(hsys);
     _&cyArrow._when  = trim(when);


%*-- common vars to draw function ---------------------------------*;
     color = %unquote(&color.); /* codes only */
     hsys  = %unquote(&hsys.);  /* for size      */
     line  = %unquote(&line.);  /* 1, 2, ..., 46 */
     size  = %unquote(&size.);  /* in hsys unit  */
     when  = %unquote(&when.);  /* draw before/after the proc output */


%*-- draw "shaft" -------------------------------------------------*;
     function = "move";                                                    (5)
          x = _&cyArrow._x1;
          y = _&cyArrow._y1;
     output;
     function = "draw";
          x     = _&cyArrow._x2;
          y     = _&cyArrow._y2;
     output;


%*-- draw "barbs" -------------------------------------------------*;
     %do i = 1 %to 2;
         function = "move";
             x = _&cyArrow._barbX&i.;
             y = _&cyArrow._barbY&i.;
         output;
         function = "draw";
             x     = _&cyArrow._x2;
             y     = _&cyArrow._y2;
         output;
     %end;


%*-- restore saved values -----------------------------------------*;
     function = "pop";                                                     (6)
     output;
     color = _&cyArrow._color;
     line  = _&cyArrow._line;
     size  = _&cyArrow._size;
     hsys  = _&cyArrow._hsys;
     when  = _&cyArrow._when;


%*-- exits --------------------------------------------------------*;
     %exit:;
     _&cyArrow._exit:;                                                     (7)
          drop _&cyArrow._:;


%mend  cyArrow;
```

SUMMARY

We have presented a macro, %CYARROW, to be used with %ANNOMAC in SAS/GRAPH, along with several examples. %CYARROW draws simple, correct arrows. It honors most of the options that work with the DRAW Annotate function (COLOR, HSYS, LINE, SIZE, and WHEN), it also has sophisticated arrow shape control options such as the length and angle of the arrow tip and its aspect ratio.

REFERENCES

Carpenter, Art (1999a). *Annotate: Simply the Basics.* BBU Press (SAS). ISBN: 1-58025-578-7. A companion web page at: <http://www.sas.com/apps/pubscat/bookdetails.jsp?pc=57320>.

Carpenter, Art (1999b). "Using ANNOTATE MACROS as Shortcuts." Information Visualization section paper. Paper 168. *SUGI 24 Proceedings.* Archived at <http://www2.sas.com/proceedings/sugi24/Infovis/p168-24.pdf>.

Dias, Carlos Tadeu Santos (2004). "arrows." A SAS-L posting on Jun. 23, 2004. Archived at <http://listserv.uga.edu/cgi-bin/wa?A2=ind0406D&L=sas-l&D=0&H=0&O=T&T=1&P=21550>.

Huang, Ya (2004). "Re: arrows." A SAS-L posting on Jun. 23, 2004. Archived at <http://listserv.uga.edu/cgi-bin/wa?A2=ind0406D&L=sas-l&D=0&H=0&O=T&T=1&P=23531>.

Rorie, Deena and Lynette Duncan (2003). "A Handy Use of the %LINE Annotate Macro." A Coders' Corner paper. Paper 83-28. *SUGI 28 Proceedings*. Archived at <http://www2.sas.com/proceedings/sugi28/083-28.pdf>.

Whitlock, Ian (2002). "Re: GOTO and LINK vs IF THEN DO." A SAS-L posting on Jul. 24, 2002. Archived at <http://listserv.uga.edu/cgi-bin/wa?A2=ind0207D&L=sas-l&D=0&H=0&O=T&T=1&m=106219&P=28195>.

ACKNOWLEDGEMENTS

We would like to thank Carlos Tadeu Santos Dias for posting an interesting question on SAS-L (Dias(2004)). Ya(2004) replied with a quick solution, which is further developed into %CYARROW that we present here.

CONTACT INFORMATION

Chang Y. Chung
Princeton University
#216 Wallace Hall
Princeton, NJ 08540
(609) 258-2360
cchung@princeton.edu
http://changchung.com/

Ya Huang
Amylin Pharmaceuticals, Inc
9373 Towne Centre Drive
San Diego, CA 92121
(858) 642-7288
yhuang@amylin.com